DWT: filters and matrices

In[148]:=
```
MyPlot[data_, range_, options_] :=
 Plot[data, range, Evaluate[options], ImageSize → Medium]
```

In[306]:=
```
fa = Filling → Axis;
pa = PlotRange → Full;
ta = TableAlignments → ".";
```

Filters

Modification of *Mathematica* data according to our conventions

In[121]:=
```
modify[fil_] := Map[ReplacePart[#, 2 -> Sqrt[2] * #[[2]]] &, fil]
```

Constructing the dual filter

In[122]:=
```
dualize[fil_] := Module[{len},
   len = Length[fil];
   Table[{k - 1, fil[[len - k + 1]][[2]] (-1) ^ (k - 1)}, {k, 1, len}]
  ]
```

Haar filter

In[309]:=
```
haarh = modify[WaveletFilterCoefficients[HaarWavelet[], "PrimalLowpass"]];
TableForm[haarh, ta]
```

Out[310]//TableForm=

0	0.707107
1	0.707107

In[311]:=
```
haarg = dualize[haarh];
TableForm[haarg, ta]
```

Out[312]//TableForm=

0	0.707107
1	-0.707107

Plotting the scaling and the wavelet function

In[149]:=
```
GraphicsGrid[{{
    MyPlot[WaveletPhi[HaarWavelet[], t], {t, -1, 2}, {fa, pa, PlotLabel → "D2-ϕ"}],
    MyPlot[WaveletPsi[HaarWavelet[], t],
     {t, -1, 2}, {fa, pa, PlotLabel → "D2-ψ"}]}}
]
```

Out[149]=



Daubechies filter D4

In[313]:=
```
daub4h =
  modify[WaveletFilterCoefficients[DaubechiesWavelet[2], "PrimalLowpass"]];
TableForm[daub4h, ta]
```

Out[314]//TableForm=

0	0.482963
1	0.836516
2	0.224144
3	-0.12941

In[315]:=
```
daub4g = dualize[daub4h];
TableForm[daub4g, ta]
```

Out[316]//TableForm=

0	-0.12941
1	-0.224144
2	0.836516
3	-0.482963

Plotting the scaling and the wavelet function

In[150]:=
```
GraphicsGrid[{{
    MyPlot[WaveletPhi[DaubechiesWavelet[2], t],
     {t, -1, 3}, {fa, pa, PlotLabel → "D4-ϕ"}],
    MyPlot[WaveletPsi[DaubechiesWavelet[2], t], {t, -1, 2},
     {fa, pa, PlotLabel → "D4-ψ"}]}}
]
```

Daubechies filter D6

In[317]:=
```
daub6h =
 modify[WaveletFilterCoefficients[DaubechiesWavelet[3], "PrimalLowpass"]];
TableForm[daub6h, ta]
```

Out[317]//TableForm=

0	0.332671
1	0.806892
2	0.459878
3	-0.135011
4	-0.0854413
5	0.0352263

Plotting the scaling and the wavelet function

In[151]:=
```
GraphicsGrid[{{
   MyPlot[WaveletPhi[DaubechiesWavelet[3], t],
    {t, -1, 4}, {fa, pa, PlotLabel → "D6-ϕ"}],
   MyPlot[WaveletPsi[DaubechiesWavelet[3], t], {t, -2, 3},
    {fa, pa, PlotLabel → "D6-ψ"}]}}
]
```

Out[151]=



Daubechies filter D20

In[318]:=
```
daub20h =
  modify[WaveletFilterCoefficients[DaubechiesWavelet[10], "PrimalLowpass"]];
TableForm[daub20h, ta]
```

Out[319]//TableForm=

0	0.0266701
1	0.188177
2	0.527201
3	0.688459
4	0.281172
5	-0.249846
6	-0.195946
7	0.127369
8	0.0930574
9	-0.0713941
10	-0.0294575
11	0.0332127
12	0.00360655
13	-0.0107332
14	0.00139535
15	0.00199241
16	-0.000685857
17	-0.000116467
18	0.0000935887
19	-0.0000132642

Plotting the scaling and the wavelet function

In[152]:=
```
GraphicsGrid[{{
   MyPlot[WaveletPhi[DaubechiesWavelet[10], t],
    {t, -1, 10}, {fa, pa, PlotLabel → "D20-ϕ"}],
   MyPlot[WaveletPsi[DaubechiesWavelet[10], t],
    {t, -5, 6}, {fa, pa, PlotLabel → "D20-ψ"}]}}
]
```

Out[152]=





Coiflet filter C6

In[320]:=
```
coif6h = modify[WaveletFilterCoefficients[CoifletWavelet[1], "PrimalLowpass"]];
TableForm[coif6h, ta]
```

Out[321]//TableForm=

-2	-0.0727326
-1	0.337898
0	0.852572
1	0.384865
2	-0.0727326
3	-0.0156557

Plotting the scaling and the wavelet function

In[155]:=
```
GraphicsGrid[{{
   MyPlot[WaveletPhi[CoifletWavelet[1], t],
    {t, -2, 2}, {fa, pa, PlotLabel → "C6-ϕ"}],
   MyPlot[WaveletPsi[CoifletWavelet[1], t], {t, -2, 3},
    {fa, pa, PlotLabel → "C6-ψ"}]}}
]
```

Out[155]=



Coiflet filter C12

In[322]:=
```
coif12h = modify[WaveletFilterCoefficients[CoifletWavelet[2], "PrimalLowpass"]];
TableForm[coif12h, ta]
```

Out[323]//TableForm=

-4	0.0163873
-3	-0.0414649
-2	-0.0673726
-1	0.38611
0	0.812724
1	0.417005
2	-0.0764886
3	-0.0594344
4	0.0236802
5	0.00561143
6	-0.00182321
7	-0.000720549

Plotting the scaling and the wavelet function

In[158]:=
```
GraphicsGrid[{{
    MyPlot[WaveletPhi[CoifletWavelet[2], t],
     {t, -3, 3}, {fa, pa, PlotLabel → "C12-ϕ"}],
    MyPlot[WaveletPsi[CoifletWavelet[2], t], {t, -3, 3},
     {fa, pa, PlotLabel → "C12-ψ"}]}}
]
```

Out[158]=



Symlet filter S6

In[324]:=
```
sym6h = modify[WaveletFilterCoefficients[SymletWavelet[3], "PrimalLowpass"]];
TableForm[sym6h, ta]
```

Out[325]//TableForm=

0	0.332671
1	0.806892
2	0.459878
3	-0.135011
4	-0.0854413
5	0.0352263

Plotting the scaling and the wavelet function

In[218]:=
```
GraphicsGrid[{{
    MyPlot[WaveletPhi[SymletWavelet[3], t],
     {t, -1, 6}, {fa, pa, PlotLabel → "S4-ϕ"}],
    MyPlot[WaveletPsi[SymletWavelet[3], t], {t, -3, 5},
     {fa, pa, PlotLabel → "S4-ψ"}]}}
]
```

Out[218]=



H and G transform matrices

In[164]:=
```
H[fil_, N_, k_] := Module[{hvec},
  hvec = PadRight[Map[#[[2]] &, fil], N];
  If[Not[Divisible[N, 2^k]], Throw["k too large"]];
  If[N < (Length[fil] - 1) * (2^k - 1), Print["wrapping!!"]];
  If[k == 1, Return[
    Table[RotateRight[hvec, 2 (m - 1)], {m, 1, N/2}]]];
  H[fil, N/(2^(k-1)), 1].H[fil, N, k - 1]
 ]
```

The H-matrix for D4 for (N,k)=(6,1)

In[165]:=
```
H[daub4h, 6, 1] // MatrixForm
```

Out[165]//MatrixForm=

0.482963	0.836516	0.224144	-0.12941	0	0
0	0	0.482963	0.836516	0.224144	-0.12941
0.224144	-0.12941	0	0	0.482963	0.836516

Orthogonality of this matrix

In[166]:=
```
Chop[H[daub4h, 6, 1] .Transpose[H[daub4h, 6, 1] ]] // MatrixForm
```

Out[166]//MatrixForm=

1.	0	0
0	1.	0
0	0	1.

The H-matrix for D4 for (N,k)=(12,2) with rounded coefficients

In[167]:=
```
Map[Round[#, 0.01] &, H[daub4h, 12, 2]] // MatrixForm
```
Out[167]//MatrixForm=

0.23	0.4	0.51	0.64	0.3	0.08	-0.01	-0.14	-0.03	0.02	0.	0.
-0.03	0.02	0.	0.	0.23	0.4	0.51	0.64	0.3	0.08	-0.01	-0.14
0.3	0.08	-0.01	-0.14	-0.03	0.02	0.	0.	0.23	0.4	0.51	0.64

Orthogonality of this matrix

In[168]:=
```
Chop[H[daub4h, 12, 2] .Transpose[H[daub4h, 12, 2] ]] // MatrixForm
```
Out[168]//MatrixForm=

1.	0	0
0	1.	0
0	0	1.

In[170]:=
```
G[fil_, N_, k_] := Module[{hvec, gvec},
  hvec = PadRight[Map[#[[2]] &, fil], N];
  gvec = PadRight[Map[#[[2]] &, dualize[fil]], N];
  If[Not[Divisible[N, 2^k]], Throw["k too large"]];
  If[N <= (Length[fil] - 1) * (2^k - 1), Print["wrapping!!"]];
  If[k == 1, Return[
    Table[RotateRight[gvec, 2 (m - 1)], {m, 1, N/2}]]];
  G[fil, N/(2^(k - 1)), 1].H[fil, N, k - 1]
 ]
```

In[171]:=
```
G[daub4h, 6, 1] // MatrixForm
```
Out[171]//MatrixForm=

-0.12941	-0.224144	0.836516	-0.482963	0	0
0	0	-0.12941	-0.224144	0.836516	-0.482963
0.836516	-0.482963	0	0	-0.12941	-0.224144

In[172]:=
```
Map[Round[#, 0.01] &, G[daub4h, 12, 2]] // MatrixForm
```
Out[172]//MatrixForm=

-0.06	-0.11	-0.14	-0.17	0.35	0.73	-0.05	-0.51	-0.11	0.06	0.	0.
-0.11	0.06	0.	0.	-0.06	-0.11	-0.14	-0.17	0.35	0.73	-0.05	-0.51
0.35	0.73	-0.05	-0.51	-0.11	0.06	0.	0.	-0.06	-0.11	-0.14	-0.17

W transform matrices

In[173]:=
```
W[fil_, N_, k_] := Module[{},
  If[Not[Divisible[N, 2^k]], Throw["k too large"]];
  If[N <= (Length[fil] - 1) * (2^k - 1), Print["wrapping!!"]];
  If[k == 1, Return[Join[H[fil, N, 1], G[fil, N, 1]]]];
  Flatten[Prepend[Table[G[fil, N, k - j], {j, 0, k - 1}], H[fil, N, k]], 1]
 ]
```

W-matrix for D4 with (N,k)=(6,1)

In[174]:=
```
Map[Round[#, 0.01] &, W[daub4h, 6, 1]] // MatrixForm
```
Out[174]//MatrixForm=

0.48	0.84	0.22	-0.13	0.	0.
0.	0.	0.48	0.84	0.22	-0.13
0.22	-0.13	0.	0.	0.48	0.84
-0.13	-0.22	0.84	-0.48	0.	0.
0.	0.	-0.13	-0.22	0.84	-0.48
0.84	-0.48	0.	0.	-0.13	-0.22

Orthogonality of this matrix

In[175]:= `Chop[W[daub4h, 6, 1].Transpose[W[daub4h, 6, 1]]] // MatrixForm`

Out[175]//MatrixForm=

1.	0	0	0	0	0
0	1.	0	0	0	0
0	0	1.	0	0	0
0	0	0	1.	0	0
0	0	0	0	1.	0
0	0	0	0	0	1.

W-matrix for D4 with (N,k)=(12,2)

In[176]:= `Map[Round[#, 0.01] &, W[daub4h, 12, 2]] // MatrixForm`

Out[176]//MatrixForm=

0.23	0.4	0.51	0.64	0.3	0.08	-0.01	-0.14	-0.03	0.02	0.	0.
-0.03	0.02	0.	0.	0.23	0.4	0.51	0.64	0.3	0.08	-0.01	-0.14
0.3	0.08	-0.01	-0.14	-0.03	0.02	0.	0.	0.23	0.4	0.51	0.64
-0.06	-0.11	-0.14	-0.17	0.35	0.73	-0.05	-0.51	-0.11	0.06	0.	0.
-0.11	0.06	0.	0.	-0.06	-0.11	-0.14	-0.17	0.35	0.73	-0.05	-0.51
0.35	0.73	-0.05	-0.51	-0.11	0.06	0.	0.	-0.06	-0.11	-0.14	-0.17
-0.13	-0.22	0.84	-0.48	0.	0.	0.	0.	0.	0.	0.	0.
0.	0.	-0.13	-0.22	0.84	-0.48	0.	0.	0.	0.	0.	0.
0.	0.	0.	0.	-0.13	-0.22	0.84	-0.48	0.	0.	0.	0.
0.	0.	0.	0.	0.	0.	-0.13	-0.22	0.84	-0.48	0.	0.
0.	0.	0.	0.	0.	0.	0.	0.	-0.13	-0.22	0.84	-0.48
0.84	-0.48	0.	0.	0.	0.	0.	0.	0.	0.	-0.13	-0.22

Orthgonality of this matrix

In[177]:= `Chop[W[daub4h, 12, 2].Transpose[W[daub4h, 12, 2]]] // MatrixForm`

Out[177]//MatrixForm=

1.	0	0	0	0	0	0	0	0	0	0	0
0	1.	0	0	0	0	0	0	0	0	0	0
0	0	1.	0	0	0	0	0	0	0	0	0
0	0	0	1.	0	0	0	0	0	0	0	0
0	0	0	0	1.	0	0	0	0	0	0	0
0	0	0	0	0	1.	0	0	0	0	0	0
0	0	0	0	0	0	1.	0	0	0	0	0
0	0	0	0	0	0	0	1.	0	0	0	0
0	0	0	0	0	0	0	0	1.	0	0	0
0	0	0	0	0	0	0	0	0	1.	0	0
0	0	0	0	0	0	0	0	0	0	1.	0
0	0	0	0	0	0	0	0	0	0	0	1.

P and Q projection matrices

In[178]:= `P[fil_, N_, k_] := Transpose[H[fil, N, k]].H[fil, N, k]`

In[179]:= `Q[fil_, N_, k_] := Transpose[G[fil, N, k]].G[fil, N, k]`

P-matrix for D4 and (N,k) = (6,1)

In[180]:= `Map[Round[#, 0.01] &, P[daub4h, 6, 1]] // MatrixForm`

Out[180]//MatrixForm=

0.28	0.38	0.11	-0.06	0.11	0.19
0.38	0.72	0.19	-0.11	-0.06	-0.11
0.11	0.19	0.28	0.38	0.11	-0.06
-0.06	-0.11	0.38	0.72	0.19	-0.11
0.11	-0.06	0.11	0.19	0.28	0.38
0.19	-0.11	-0.06	-0.11	0.38	0.72

Idempotency $P^2=P$ of this matrix

In[181]:=
```
Chop[P[daub4h, 6, 1].P[daub4h, 6, 1] - P[daub4h, 6, 1]] // MatrixForm
```

Out[181]//MatrixForm=

0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0

P-matrix for D4 and (N,k) = (12,2)

In[182]:=
```
Map[Round[#, 0.01] &, P[daub4h, 12, 2]] // MatrixForm
```

Out[182]//MatrixForm=

0.14	0.12	0.12	0.11	0.05	0.01	-0.02	-0.05	0.05	0.12	0.15	0.19
0.12	0.17	0.21	0.25	0.12	0.04	0.	-0.04	0.01	0.04	0.04	0.05
0.12	0.21	0.26	0.33	0.15	0.04	-0.01	-0.07	-0.02	0.	-0.01	-0.01
0.11	0.25	0.33	0.42	0.19	0.05	-0.01	-0.09	-0.05	-0.04	-0.07	-0.09
0.05	0.12	0.15	0.19	0.14	0.12	0.12	0.11	0.05	0.01	-0.02	-0.05
0.01	0.04	0.04	0.05	0.12	0.17	0.21	0.25	0.12	0.04	0.	-0.04
-0.02	0.	-0.01	-0.01	0.12	0.21	0.26	0.33	0.15	0.04	-0.01	-0.07
-0.05	-0.04	-0.07	-0.09	0.11	0.25	0.33	0.42	0.19	0.05	-0.01	-0.09
0.05	0.01	-0.02	-0.05	0.05	0.12	0.15	0.19	0.14	0.12	0.12	0.11
0.12	0.04	0.	-0.04	0.01	0.04	0.04	0.05	0.12	0.17	0.21	0.25
0.15	0.04	-0.01	-0.07	-0.02	0.	-0.01	-0.01	0.12	0.21	0.26	0.33
0.19	0.05	-0.01	-0.09	-0.05	-0.04	-0.07	-0.09	0.11	0.25	0.33	0.42

Idempotency of this matrix

In[183]:=
```
Chop[P[daub4h, 12, 2].P[daub4h, 12, 2] - P[daub4h, 12, 2]] // MatrixForm
```

Out[183]//MatrixForm=

0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0

Q-matrix for D4 and (N,k) = (6,1)

In[184]:=
```
Map[Round[#, 0.01] &, Q[daub4h, 6, 1]] // MatrixForm
```

Out[184]//MatrixForm=

0.72	-0.38	-0.11	0.06	-0.11	-0.19
-0.38	0.28	-0.19	0.11	0.06	0.11
-0.11	-0.19	0.72	-0.38	-0.11	0.06
0.06	0.11	-0.38	0.28	-0.19	0.11
-0.11	0.06	-0.11	-0.19	0.72	-0.38
-0.19	0.11	0.06	0.11	-0.38	0.28

Idempotency of this matrix

In[185]:= `Chop[P[daub4h, 6, 1].P[daub4h, 6, 1] - P[daub4h, 6, 1]] // MatrixForm`

Out[185]//MatrixForm=

0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0

Q-matrix for D4 and (N,k) = (12,2)

In[186]:= `Map[Round[#, 0.01] &, Q[daub4h, 12, 2]] // MatrixForm`

Out[186]//MatrixForm=

0.14	0.26	-0.01	-0.17	-0.05	-0.01	0.02	0.05	-0.05	-0.12	-0.04	0.
0.26	0.55	-0.02	-0.35	-0.12	-0.04	0.	0.04	-0.01	-0.04	-0.1	-0.16
-0.01	-0.02	0.02	0.05	-0.04	-0.1	0.01	0.07	0.02	0.	0.01	0.01
-0.17	-0.35	0.05	0.29	0.	-0.16	0.01	0.09	0.05	0.04	0.07	0.09
-0.05	-0.12	-0.04	0.	0.14	0.26	-0.01	-0.17	-0.05	-0.01	0.02	0.05
-0.01	-0.04	-0.1	-0.16	0.26	0.55	-0.02	-0.35	-0.12	-0.04	0.	0.04
0.02	0.	0.01	0.01	-0.01	-0.02	0.02	0.05	-0.04	-0.1	0.01	0.07
0.05	0.04	0.07	0.09	-0.17	-0.35	0.05	0.29	0.	-0.16	0.01	0.09
-0.05	-0.01	0.02	0.05	-0.05	-0.12	-0.04	0.	0.14	0.26	-0.01	-0.17
-0.12	-0.04	0.	0.04	-0.01	-0.04	-0.1	-0.16	0.26	0.55	-0.02	-0.35
-0.04	-0.1	0.01	0.07	0.02	0.	0.01	0.01	-0.01	-0.02	0.02	0.05
0.	-0.16	0.01	0.09	0.05	0.04	0.07	0.09	-0.17	-0.35	0.05	0.29

Idempotency of this matrix

In[187]:= `Chop[Q[daub4h, 12, 2].Q[daub4h, 12, 2] - Q[daub4h, 12, 2]] // MatrixForm`

Out[187]//MatrixForm=

0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0

$P_{N,k}.Q_{N,k} = 0$ for D4 and $(N, k) = (6, 1)$

In[188]:= `Chop[P[daub4h, 6, 1].Q[daub4h, 6, 1]] // MatrixForm`

Out[188]//MatrixForm=

0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0
0	0	0	0	0	0

$Q_{N,k}.P_{N,k} = 0$ for D4 and $(N, k) = (12, 2)$

```
Chop[Q[daub4h, 12, 2].P[daub4h, 12, 2]] // MatrixForm
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

$P_{N,1} + Q_{N,1} = \mathrm{Id}$ for D4

In[189]:=
```
Chop[P[daub4h, 6, 1] + Q[daub4h, 6, 1]] // MatrixForm
```

Out[189]//MatrixForm=

$$\begin{pmatrix} 1. & 0 & 0 & 0 & 0 & 0 \\ 0 & 1. & 0 & 0 & 0 & 0 \\ 0 & 0 & 1. & 0 & 0 & 0 \\ 0 & 0 & 0 & 1. & 0 & 0 \\ 0 & 0 & 0 & 0 & 1. & 0 \\ 0 & 0 & 0 & 0 & 0 & 1. \end{pmatrix}$$

$P_{N,1} + Q_{N,1} = \mathrm{Id}$ for D6

In[190]:=
```
Chop[P[daub6h, 8, 1] + Q[daub6h, 8, 1]] // MatrixForm
```

Out[190]//MatrixForm=

$$\begin{pmatrix} 1. & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1. & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1. & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1. & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1. & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1. & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1. & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1. \end{pmatrix}$$

Approximation of ECG-data

ECG-data

In[191]:=
```
seq = Import[
  "~/LEHRE/Wavelets-All/WTBV-13/Daten/heart.dat"];
seq = Flatten[seq];
```

In[192]:=
```
ListLinePlot[{seq, P[haarh, 2048, 5].seq}]
```

Out[192]=


x

In[193]:=
```
showProj[data_, wav_, k_] := Module[{N, Detail, Approx},
  N = Length[data];
  If[Not[Divisible[N, 2^k]], Throw["k too large"]];
  Do[
   Detail[j] = Q[wav, N, j].data;
   d[j] = ListLinePlot[Detail[j],
     PlotRange → {-1.5, 1.5},
     ImageSize → {500, 100},
     AspectRatio → 1 / 5,
      PlotLabel → {"Detail Level", j}], {j, 1, k}];
  Approx[k] = P[wav, N, k].data;
  a[k] = ListLinePlot[Approx[k],
    PlotRange → {-1.5, 1.5},
    ImageSize → {400, 100},
    AspectRatio → 1 / 5, PlotLabel → {"Approximation Level", k}];
  GraphicsGrid[Append[Table[{d[j]}, {j, 1, 6}], {a[k]}]]
 ]
```

Wavlelet decomposition using a 6-level transformation

Haar wavelet

In[194]:= showProj[seq, haarh, 6]

Out[194]=



Daubechies-4 wavelet

In[195]:=
```
showProj[seq, daub4h, 6]
```

Out[195]=



Symlet-6 wavelet

In[219]:= `showProj[seq, sym6h, 6]`

Out[219]=



Comparing approximations on level 6

In[220]:=
```
compareProj[data_, wav_, k_] := Module[{N, L, filter, proj, plot},
  N = Length[data];
  If[Not[Divisible[N, 2^k]], Throw["k too large"]];
  L = Length[wav];
  Do[filter[n] = ToLowerCase[ToString[wav[[n]]]], {n, 1, L}];
  ft = Table[filter[n], {n, 1, L}];
  ft = Map[ToExpression[#] &, ft];
  proj = Map[P[#, N, k].data &, ft];
  Do[plot[n] = ListLinePlot[proj[[n]], PlotLabel → wav[[n]]], {n, 1, L}];
  GraphicsGrid[
   Partition[Prepend[Table[plot[n], {n, 1, L}], ListLinePlot[data]], 2, 2, 1, {}]]
 ]
```

In[225]:=
```
compareProj[seq, {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 6]
```

Out[225]=



Comparing approximations on level 5

In[226]:=
```
compareProj[seq, {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 5]
```

Out[226]=



Comparing approximations on level 4

In[227]:=
```
compareProj[seq, {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 4]
```

Out[227]=



Approximating a polynomial function

In[228]:=
```
Clear[poly]
```

In[261]:=
```
poly[t_] := t^5 - 2 t^2 + t
```

In[262]:=
```
Plot[poly[t], {t, 0, 1}]
```

Out[262]=



In[263]:=
```
polylist = Table[poly[t], {t, 0, 1, 2^(-11)}];
```

In[264]:=
```
polylist = Delete[polylist, {2049}];
```

In[265]:=
```
Length[polylist]
```
Out[265]=
```
2048
```

Comparing approximations on level 4

In[266]:=
```
compareProj[polylist, {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 4]
```
Out[266]=



Comparing approximations on level 6

In[327]:=
```
compareProj[polylist, {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 6]
```

Out[327]=



Direct comparison

In[259]:=
```
showProj[data_, wav_, k_, range_] := Module[{N, L, filter, ftab, ft, proj, plot},
  N = Length[data];
  If[Not[Divisible[N, 2^k]], Throw["k too large"]];
  L = Length[wav];
  {low, high} = range;
  Do[filter[n] = ToLowerCase[ToString[wav[[n]]]], {n, 1, L}];
  ftab = Table[filter[n], {n, 1, L}];
  ft = Map[ToExpression[#] &, ftab];
  proj = Map[P[#, N, k].data &, ft];
  ListLinePlot[
   Append[Table[proj[[n]][[low ;; high]], {n, 1, L}], data[[low ;; high]]],
   PlotRange → {{low, high}, Full}, DataRange → {low, high},
   PlotLabels → ftab, PlotStyle → Join[Table[Dashed, {l, 1, L}], {Black}]]
 ]
```

Level 6

In[268]:=
```
showProj[polylist,
 {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 6, {1, 2048}]
```

Out[268]=



In[286]:=
```
showProj[polylist,
 {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 6, {400, 600}]
```

Out[286]=



In[272]:=
```
showProj[polylist,
 {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 6, {680, 720}]
```

Out[272]=



In[273]:= `showProj[polylist, {Daub20h, Coif12h, Sym6h}, 6, {600, 800}]`

Out[273]=



In[274]:= `showProj[polylist, {Coif12h}, 7, {1, 2048}]`

Out[274]=



In[277]:=
```
showProj[polylist,
  {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 6, {2030, 2048}]
```

Out[277]=



Level 5

In[280]:=
```
showProj[polylist,
  {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 5, {2030, 2048}]
```

Out[280]=



Level 4

In[285]:=
```
showProj[polylist,
  {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 4, {400, 600}]
```

Out[285]=



Level 3

In[289]:=
```
showProj[polylist,
  {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 3, {400, 600}]
```

Out[289]=



In[288]:=
```
showProj[polylist,
  {Haarh, Daub4h, Daub6h, Daub20h, Coif6h, Coif12h, Sym6h}, 3, {500, 550}]
```

Out[288]=